

Example 21 - Age and Memory

DATA LIST FREE / age mem.

BEGIN DATA

1 36 1 32 1 29 1 31 1 28 1 34 1 32 1 27 1 34 1 30 1 26 1 31
2 26 2 30 2 28 2 31 2 31 2 24 2 27 2 29 2 32 2 26 2 30 2 25

END DATA.

*On average, younger adults remember 32 words on this task. Do results suggest that this sample comes from a population that remembers more or fewer words.

TTEST TESTVALUE = 32 / VARI = mem.

	N	Mean	Std. Deviation	Std. Error Mean
mem	24	29.54167	3.064228	.625483

Test Value = 32				
	t	df	Sig. (2-tailed)	Mean Difference
mem	-3.930	23	.001	-2.458333

COMPUTE mem32 = mem-32.

GLM mem32 /PRINT = DESCR.

Dependent Variable: mem32

	Mean	Std. Deviation	N
	-2.45833	3.064228	24

Dependent Variable: mem32

Source	Type III Sum of Squares	df	Mean Square	F	Sig.
Corrected Model	.000	0	.	.	.
Intercept	145.042	1	145.042	15.45	.001
Error	215.958	23	9.389		
Total	361.000	24			
Corrected Total	215.958	23			

*Sampling Distribution: 100,000 samples, mu = 32, sigma = 5, n = 24.

SET SEED = 14222318.

INPUT PROGRAM.

LOOP sample = 1 TO 100000.

DO REPEAT y = y1 TO y24.

COMPUTE #a = RV.NORM(0,1).


COMPUTE y = 32+5*(-.707107*#a + .707107*RV.NORM(0,1)).

END REPEAT.
END CASE.
END LOOP.
END FILE.
END INPUT PROGRAM.

COMPUTE mean = MEAN(y1 TO y24).
COMPUTE variance = SD(y1 TO y24)**2.
COMPUTE t = (mean-32)/SQRT(variance/24).
COMPUTE pobs = ABS(t) GE 3.930297.
COMPUTE palpha = ABS(t) GE 2.069.
MEANS pobs palpha.

	pobs	palpha
Mean	.00073	.04863

FREQ t /FORM = NOTABLE /HISTOGRAM.



*Independent groups.
TTEST VARI = mem /GROUP = age.

	age	N	Mean	Std. Deviation
mem	1.000	12	30.83333	3.010084
	2.000	12	28.25000	2.632835

t-test for Equality of Means					
	t	df	Sig. (2-tailed)	Mean Difference	Std. Error Difference
mem	2.238	22	.036	2.583333	1.154427

GLM mem BY age /PRINT = DESCR.

Dependent Variable: mem


age	Mean	Std. Deviation	N
1.000	30.83333	3.010084	12
2.000	28.25000	2.632835	12
Total	29.54167	3.064228	24

Dependent Variable: mem

Source	Type III Sum of Squares	df	Mean Square	F	Sig.
age	40.042	1	40.042	5.008	.036
Error	175.917	22	7.996		
Corrected Total	215.958	23			

```
*Simulation H0 true: mu1 = mu2.
SET SEED = 14222318.
INPUT PROGRAM.
LOOP sample = 1 TO 100000.
DO REPEAT y = y1 TO y12.
  COMPUTE y = 32+5*RV.NORM(0,1).
END REPEAT.
DO REPEAT o = o1 TO o12.
  COMPUTE o = 32+5*RV.NORM(0,1).
END REPEAT.
END CASE.
END LOOP.
END FILE.
END INPUT PROGRAM.
COMPUTE diff = MEAN(y1 TO y12) - MEAN(o1 TO o12).
COMPUTE var = (SD(y1 TO y12)**2 + SD(o1 TO o12)**2)/2.
COMPUTE t = diff/SQRT(var*(1/12 + 1/12)).

FREQ t /FORMAT = NOTABLE /HIST.
```



```
COMPUTE pobsnon = ABS(t) GE 2.238.
COMPUTE pobsdir = t GE 2.238.
COMPUTE palphanon = ABS(t) GE 2.074.
COMPUTE palphadir = t GE 1.717.
MEANS pobsnon pobsdir palphanon palphadir.
```

	pobsnon	pobsdir	palphanon	palphadir
Mean	.03564	.01735	.05030	.04984

```
*SPSS commands to generate sample data.
SET SEED = 14222318.
INPUT PROGRAM.
LOOP #s = 1 TO 24.
END CASE.
END LOOP.
END FILE.
END INPUT PROGRAM.
COMPUTE #a = RV.NORM(0,1).
COMPUTE yrs = RND(55+10*#a).
COMPUTE mem = RND(30+5*(-.7071*#a + .7071*RV.NORM(0,1))).
SORT CASES BY yrs.
RANK yrs /NFILE(2) INTO age.
FORMAT mem yrs (F2.0) age (F1.0).
DELETE VARI yrs.
LIST age mem.
```